

```

// ++++++
funcprot(0);
// ++++++



function [x, y, z]=fSphereData()
    u=linspace(-%pi/2, %pi/2, 100);
    v=linspace(0, 2*%pi, 100);

    x=cos(u)'.*cos(v);
    y=cos(u)'.*sin(v);
    z=sin(u)'.*ones(v);
endfunction

// ++++++



// defines the color index for each facet, considering dot product
// of each direction of the light (cosang)

function [col]=setSphereColor(cosang, icol)
    c1=-0.03; // -88.3°
    c2= 0.03; // +88.3°
    c3=1;

    col=zeros(cosang);

    I=find(cosang<=c1);
    col(I)=icol(1);
    // dark zone

    I=find(cosang>c1 & cosang<c2);
    col(I)=icol(1)+round(((cosang(I)-c1)/(c2-c1))*(icol(2)-icol(1)));
    // small middle zone, uses interpolation

    I=find(cosang>=c2);
    col(I)=icol(2)+round(((cosang(I)-c2)/(c3-c2))*(icol(3)-icol(2)));
    // yellow zone, uses interpolation

endfunction

// ++++++



// Defines 63 colors

function [Cmap0, Cmap1, Cmap2]=setSphereColorMap()
    c0=0.3*[1,1,1]; // dark grey
    c1=[1,0.85,0.1]; // dark yellow
    c2=[1,1,0.5]; // light yellow

    Cmap0=[1,1,1]; // 1 color

    p=(0:30)'/30; // 31 values

    Cmap1=CL_dMult(1-p, c0)+CL_dMult(p, c1);
    // grey .. dark yellow, 31 colors

```

```

Cmap2=CL_dMult(1-p, c1)+CL_dMult(p, c2);
// dark yellow .. light yellow, 31 colors
endfunction

// ++++++
// Test

[x,y,z]=fSphereData();
[Cmap0,Cmap1,Cmap2]=setSphereColorMap();

tic();
f=scf();
clf(f);

f.visible="off";
f.immediate_drawing="off";

f.color_map=[Cmap0;Cmap1;Cmap2];
icol=1+[1,31,62]; // Color indices

plot3d2(x,y,z);

e=gce();
fcData=e.data;

// The light comes from +z
// d.z is the scalar product of the vector with the direction
// of the light => color

icol=1+[1,31,62];
sphereCol=setSphereColor(fcData.z,icol);

// ++++++
// Rotation matrix

RotM=[ 0.4344354    0.3096681   -0.8457964
      0.8327448    0.2197391    0.5081838
      0.3432229   -0.9251056   -0.1624121];

dim = size(fcData.x); // 4.      9801.

coord = RotM * [fcData.x(:)'; fcData.y(:)'; fcData.z(:)'];
// size(coord)) => 3.      39204.

x = matrix(coord(1,:), dim); // size(x) => 4.      9801.
y = matrix(coord(2,:), dim);
z = matrix(coord(3,:), dim);

// ++++++
TL=tlist(["3d" "x" "y" "z" "color"],x,y,z,sphereCol);

```

```
e.data=TL;
e.color_flag=3; // 'interpolated' shading
e.color_mode=-1; // no edges

// graphic settings
a=gca();
a.axes_visible=[ "off", "off", "off" ];
a.box="off";
a.x_label.text="";
a.y_label.text="";
a.z_label.text="";
a.isoview="on";
a.rotation_angles = [ 0, -90 ];

f.immediate_drawing="on";
f.visible="on";

dt=(0.8-toe())*1000;
sleep(max(dt,1)); // sleeps at most 0.8 s
```